
Brug af dette dokument
Dette dokument er primært skrevet til Internet-udbydere, der skal oprette scriptfiler, som slutbrugere kan tilpasse 
til deres egne forbindelser. 

Sådan bruges scripting i Internet Explorer
Med IEscript.exe kan en slutbruger bruge en scriptfil (.scp) til at etablere en forbindelse. IESCRIPT tilføjer 
indstillinger i forbindelsesfilen (.con) som vist nedenfor. Brug IESCRIPT til at redigere forbindelsesfilen. Rediger 
ikke forbindelsesfilen manuelt.

Afsnitsnavn: Script

Følgende nøgleord understøttes:

ScriptEnabled
Yes/No  Angiver, om Scriptværktøj skal aktiveres. 

ScriptFileName
En komplet streng, som angiver placeringen af den scriptfil, der er associeret med denne 
forbindelsesfil.

ScriptTerminal
Yes/No  Angiver, om vinduet for scriptterminalen skal indlæses under afspilning af scriptet.

ScriptRecord
Yes/No  Angiver, om den eksekverbare scriptfil skal gå i indspilningstilstand ved aktivering.

Eksempler på nøgleord i forbindelsesfiler til scripting
Eksempel 1:

[Script]
ScriptEnabled=Yes
ScriptFileName=C:\Shiva\Myscript.Scp
ScriptTerminal=Yes
ScriptRecord=No

Eksempel 2:

[Script]
ScriptEnabled=Yes
ScriptFileName=C:\Shiva\NewScrip.Scp
ScriptTerminal=Yes



ScriptRecord=Yes

Bemærk!
· Eksempel 2 viser det, der vises, mens en scriptfil indspilles.

Oprettelse af et script i Internet Explorer
Hvis du vil oprette en scriptfil til slutbrugere, skal du først indspille en scriptfil med Scripter.exe. Når denne fil er 
oprettet, kan slutbrugeren redigere den og indsætte eget navn og egen adgangskode.

Sådan indspilles og afspilles scriptet med TTY-vinduet
1. Kør Scripter.exe. Denne fil er normalt placeret i den samme mappe som Internet Explorer-

programmet. 
2. Klik på Indspil i menuen Filer.
3. Find en scriptfil, og marker derefter en scriptfil (.scp) på fillisten. Eller skriv et nyt filnavn. 

Når indspilningen startes, vises følgende vindue.
4. Åbn din Internet Explorer-programgruppe, og vælg derefter Ny forbindelse. Følg 

instruktionerne på skærmen for at oprette en ny forbindelse. Eller vælg ikonet for en 
forbindelse, som du har oprettet tidligere. 

5. Marker ikonet for forbindelsen. Vælg Egenskaber i menuen Filer. Sørg for, at 
afkrydsningsfeltet "Vis terminalvinduet, når opkaldet er foretaget" er markeret.

6. Etabler en forbindelse til Internet-serveren.
Meddelelsen "Started!!!" vises i vinduet ShivaRemote Scripter.

7. Klik i indspilningsvinduet (der fungerer som terminalvindue i øjeblikket), og angiv derefter 
de oplysninger, der er nødvendige for at logge på Internet-udbyderens server.

8. Når du har logget på, skal du klikke på Afslut indspilning i vinduet Scripter.
9. Klik på OK i vinduet ShivaPPP Scripter.
10. Klik på Annuller i forbindelsesdialogboksen for at afslutte kommunikationen.

Bemærk!  
Når denne scriptfil bruges, skal slutbrugeren sørge for, at afkrydsningsfeltet "Vis 
terminalvinduet, når opkaldet er foretaget" ikke er markeret.

Klik på Stop, når du har angivet de nødvendige oplysninger i scriptet. Vinduet lukkes, og den 
almindelige PPP-forhandling indledes.

Der oprettes følgende script af programmet:

; C:\SCRIPTS\IE\LOGIN.SCP
; Oprettet:  7-1-1996 kl. 17:07:11
;
;
proc main
string szPassword
transmit "^M"
waitfor "Værtsnavn:  ",matchcase until 15
transmit "spry01^M"
waitfor "UIC: ",matchcase until 6
transmit "spry053514^M"
waitfor "Adgangskode: ",matchcase until 3
if $PASSWORD then
 transmit $PASSWORD



 transmit "^M"
 goto doneTxPassword
endif
getinput "Adgangskode: " szPassword
transmit szPassword
transmit "^M"
doneTxPassword:
endproc

Rediger scriptfil ud fra den indspillede fil, eller opret en scriptfil fra bunden
Ved hjælp af Scripter.exe indspilles der en scriptfil, som indeholder brugerspecifikke oplysninger, herunder 
brugernavn og adgangskode. Hvis du er Internet-udbyder, ønsker du muligvis at levere en standardscriptfil, som 
kan redigeres og bruges af alle slutbrugere.

Benyt følgende eksempel som retningslinje for oprettelsen af et standardscript, som ikke indeholder 
brugerspecifikke oplysninger.

;---------------------------------------------------
; Filnavn: LOGIN01.SCP
; Oprettet den: 10-18-1996
; Oprettet kl.: 12:00:00
;---------------------------------------------------

proc main

; Nogle systemer kræver et tastetryk for at initiere logonproceduren
;-------------------------------------------------------------------

transmit "^M"

; Send værtsnavn, hard-coded af Internet-udbyderen
;-------------------------------------------------

waitfor "Værtsnavn:"
transmit "værtsnavn"
transmit "^M"

; Send brugernavn
; Hentes fra boksen Brugernavn i den associerede forbindelsesfil
;---------------------------------------------------------------

waitfor "Brugernavn:"
transmit $USERID
transmit "^M"

; Send adgangskode



; Hentes fra boksen Adgangskode i den associerede forbindelsesfil
;----------------------------------------------------------------

waitfor "Adgangskode:"
transmit $PASSWORD
transmit "^M"

endproc

Yderligere oplysninger om Scripting 

1.0   Oversigt
Mange Internet-udbydere og onlinetjenester kræver, at slutbrugeren manuelt angiver oplysninger, f.eks. 
brugernavn og adgangskode, for et etablere forbindelse. Når scripting understøttes i forbindelse med Netværk 
via modem, kan brugeren skrive et script, som automatiserer denne proces. 

Et script er en tekstfil, som indeholder en række kommandoer, parametre og udtryk, som Internet-udbyderen 
eller onlinetjenesten kræver for, at slutbrugeren kan etablere forbindelse eller bruge tjenesten. Du kan bruge en 
teksteditor, f.eks. Microsoft Notesblok, til at oprette en scriptfil. Når du har oprettet scriptfilen, kan du derefter 
knytte den til en bestemt Netværk via modem-forbindelse ved at køre Scriptværktøj til Netværk via modem.

2.0   Den grundlæggende struktur i et script
De grundlæggende instruktioner, som en scriptfil indeholder, kaldes kommandoer. Nogle kommandoer kræver 
parametre, som præciserer, hvad kommandoerne skal udføre. Et udtryk er en kombination af operatorer og 
argumenter, som udgør et resultat. Udtryk kan bruges som værdier i alle kommandoer. Eksempler på udtryk 
omfatter matematiske operatorer, relationelle sammenligninger og sammenkædninger af strenge.

Et script til Netværk via modem benytter følgende struktur:

;
; En kommentar indledes af et semikolon og går til
; linjeslut.
;

proc main
;  Et script kan indeholde et hvilket som helst antal 
;  variabler og kommandoer

variabelerklæringer

kommandoer

endproc

Et script skal indeholde en hovedprocedure, der angives ved hjælp af nøgleordet proc og et tilsvarende 
endproc-nøgleord, der angiver slutningen af proceduren.  

Du skal oprette erklæringer for alle variabler, inden du tilføjer kommandoer. Den første kommando i 
hovedproceduren køres, og derefter køres de efterfølgende kommandoer i scriptfilen i den rækkefølge, som de 
forekommer i. Scriptet slutter samtidig med slutningen af hovedproceduren.

3.0   Variabler



Scriptfiler kan indeholde variabler. Variabelnavne skal starte med et bogstav eller et understregningstegn (_) og 
kan indeholde enhver sekvens af store eller små bogstaver, tal og understregningstegn. Du kan ikke bruge et 
reserveret ord som variabelnavn. Yderligere oplysninger finder du i listen over reserverede ord i slutningen af 
dette dokument.

Du skal oprette erklæringer for alle variabler, inden de bruges. Når du opretter en erklæring for en variabel, skal 
du også definere variablens type. En variabel af en bestemt type kan kun indeholde værdier af den pågældende 
type. Følgende tre typer variabler understøttes:

Type Heltal
Beskrivelse Et negativt eller positivt tal, f.eks. 7, -12 eller 5698.

Type Streng
Beskrivelse En række tegn i dobbelte anførselstegn, f.eks. "Hej med jer!" eller "Indtast 

adgangskode:".

Type Boolesk
Beskrivelse En logisk boolesk værdi, der er TRUE (sand) eller FALSE (falsk).

Variabler tildeles værdier ved hjælp af følgende udsagn:

variabel = udtryk

Variablen tildeles det evaluerede udtryk.  

Eksempler:

integer count = 5
integer timeout = (4 * 3)
integer i

boolean bDone = FALSE

string szIP = (getip 2)

set ipaddr szIP

Eksemplerne ovenfor bruger scriptkommandoer, som ikke understøttes af Shiva.

3.1   Systemvariabler
Systemvariabler defineres af scriptkommandoer eller de oplysninger, som du angiver, når du etablerer en 
forbindelse. Systemvariabler er skrivebeskyttede, hvilket betyder, at de ikke kan ændres i scriptet. 
Systemvariablerne er:

Navn $USERID
Type Streng
Beskrivelse Brugernavnet for den nuværende forbindelse. Værdien af denne variabel er det 

brugernavn, der er angivet dialogboksen for forbindelse i forbindelsesprogrammet.

Navn $PASSWORD 
Type Streng
Beskrivelse Adgangskoden til den nuværende forbindelse. Værdien af denne variabel er den 

adgangskode, der er angivet dialogboksen for forbindelse i forbindelsesprogrammet.



Navn $SUCCESS
Type Boolesk
Beskrivelse Denne variabel defineres af visse kommandoer for at angive, om en kommando lykkes. 

Et script kan træffe beslutninger på grundlag af værdien af denne variabel.

Navn $FAILURE
Type Boolesk
Beskrivelse Denne variabel defineres af visse kommandoer for at angive, om en kommando 

mislykkes. Et script kan træffe beslutninger på grundlag af værdien af denne variabel.

Disse variabler kan bruges, hvis der bruges et udtryk af en tilsvarende type. F.eks. er

transmit $USERID

en gyldig kommando, fordi $USERID er en variabel af typen streng.

4.0   Ordrette strenge
Scripting til Netværk via modem understøtter ESC-sekvenser og oversættelser af indsætningspunkt som 
beskrevet i det følgende.

Ordret streng Beskrivelse
Ordret streng ^tegn
Beskrivelse Oversættelse af indsætningspunkt

Hvis tegn er en værdi mellem '@' og '_', oversættes tegnsekvensen til en værdi 
mellem 0 og 31 på én byte. ^M konverteres f.eks. til linjeskift.

Hvis tegn er en værdi mellem '@' og '_', oversættes tegnsekvensen til en værdi mellem 
0 og 31 på én byte. ^M konverteres f.eks. til linjeskift.

Hvis tegn er en værdi mellem 'a' og 'z', oversættes tegnsekvensen til en værdi mellem 1 
og 26 på én byte.

Hvis tegn er en anden værdi, behandles tegnsekvensen ikke på en speciel måde.

Ordret streng <cr>
Beskrivelse Linjeskift 

Ordret streng <lf>
Beskrivelse Linjeskift

Ordret streng \"
Beskrivelse Dobbelt anførselstegn

Ordret streng \^
Beskrivelse Enkelt indsætningspunkt

Ordret streng \<
Beskrivelse Enkelt mindre end-tegn (<)

Ordret streng \\



Beskrivelse Omvendt skråstreg

Eksempler:

transmit "^M"
transmit "Joe^M"
transmit "<cr><lf>"
waitfor "<cr><lf>"

5.0  Udtryk
Et udtryk er en kombination af operatorer og argumenter, som evalueres til et resultat. Udtryk kan bruges som 
værdier i alle kommandoer.

Et udtryk kan kombinere alle variabler, heltal, strenge eller booleske værdier med de unære og binære værdier i 
følgende tabeller. Alle unære operatorer har første prioritet. Prioriteringen af binære operatorer angives af deres 
placering i tabellen. 

Følgende er de unære operatorer:

Operator -
Operationstype Unært minus

Operator !
Operationstype Komplement af en

De binære operatorer er angivet i følgende tabel i deres prioriteringsrækkefølge. Operatorer med den højeste 
prioritet vises først:

Operatorer *  /
Operationstype Multiplikation
Begrænsninger for type Heltal

Operatorer +  -
Operationstype Addition heltal
Begrænsninger for type Strenge (+ kun)

Operatorer < > <= >=
Operationstype Relationel
Begrænsninger for type Heltal

Operatorer == !=
Operationstype Lighed
Begrænsninger for type Heltal, strenge, boolesk

Operatorer and
Operationstype Logisk OG
Begrænsninger for type Boolesk

Operatorer or
Operationstype Logisk ELLER
Begrænsninger for type Boolesk

Eksempler:



count = 3 + 5 * 40
transmit "Hej" + " piger"
delay 24 / (7 - 1)

6.0   Kommentarer
Al tekst på en linje, der kommer efter et semikolon, ignoreres.

Eksempler:

; dette er en kommentar

transmit "hej" ; overfør strengen "hej"

7.0   Nøgleord
Nøgleord angiver scriptets struktur. I modsætning til kommandoer udfører de ikke en handling. Nøgleordene er 
anført nedenfor.

proc navn 

Angiver begyndelsen af en procedure. Alle scriptfiler skal indeholde en hovedprocedure (proc hoved). 
Udførelsen af scriptet starter ved hovedproceduren og standser ved slutningen af hovedproceduren.

endproc

Angiver slutningen af en procedure. Når scriptfilen køres til udsagnet endproc for hovedproceduren, 
startes PPP eller SLIP af Scripting til Netværk via Modem. 

integer navn [ = værdi ] 

Erklærer en variabel af typen heltal. Du kan bruge ethvert numerisk udtryk eller enhver variabel til at 
initialisere variablen.

string navn [ = værdi ] 

Erklærer en variabel af typen streng. Du kan bruge enhver ordret streng eller variabel til at initialisere 
variablen.

boolean navn [ = værdi ] 

Erklærer en variabel af typen boolesk. Du kan bruge ethvert boolesk udtryk eller enhver variabel til at 
initialisere variablen.

8.0   Kommandoer
Den første implementering af scriptsproget til opkaldsprogrammet Internet Explorer (Dialer) er en 
del af det kommandosprog for Scriptværktøj til Netværk via modem, der er defineret i Microsoft 
Windows(R) 95.  

Alle kommandoer er reserverede ord, hvilket betyder, at du ikke kan erklære variabler, som har de samme navne
som kommandoerne. Følgende kommandoer benyttes:

delay nSekunder

Standser det antal sekunder, der angives af nSekunder, inden den næste kommando i scriptfilen køres.



Eksempler:

delay 2 ; standser i 2 sekunder
delay x * 3 ; standser i x * 3 sekunder

goto label

Flytter til det sted i scriptet, der angives af label, og fortsætter med at køre de efterfølgende 
kommandoer.

Eksempel:

 waitfor "Prompt>" until 10
 if !$SUCCESS then

goto BailOut ; flytter til BailOut og udfører kommandoerne 
; derefter

 endif

 transmit "bbs^M"
 goto End

BailOut:
 transmit "^M"

halt

Stopper scriptet. Denne kommando fjerner ikke terminaldialogvinduet. Du skal klikke på Fortsæt for at 
etablere forbindelse. Du kan ikke genstarte scriptet.

if betingelse then
kommandoer

endif

Udfører en serie kommandoer, hvis betingelse er TRUE (sand).

Eksempel:

if $USERID == "John" then
transmit "Johnny^M"

endif

label :

Angiver det sted i scriptet, der skal flyttes til. En label skal være et entydigt navn og følge 
navnekonventionerne for variabler.

transmit streng [ , raw ]

Sender de tegn, der angives af streng, til fjerncomputeren. 

Fjerncomputeren genkender ESC-sekvenser og oversættelser af indsætningspunkt, medmindre du 



medtager parameteren raw i kommandoen. Parameteren raw kan bruges til at overføre $USERID- og 
$PASSWORD-systemvariabler, når brugernavn og adgangskode indeholder tegnsekvenser, som ville 
blive fortolket som indsætningspunkt eller ESC-sekvenser uden parameteren raw.

Eksempler:

transmit "slip" + "^M"
transmit $USERID, raw

waitfor streng [ , matchcase ] [ then label 
{ , streng [ , matchcase ] then label } ]
[ until tid ]

Venter, indtil computeren modtager en eller flere af de angivne strenge fra fjerncomputeren. Der skelnes 
ikke mellem store og små bogstaver i parameteren streng, medmindre parameteren matchcase 
medtages.

Hvis der modtages en tilsvarende streng, og parameteren then label derefter bruges, flytter denne 
kommando til det sted i scriptfilen, som label angiver.

Den valgfrie parameter until tid definerer det antal sekunder, som computeren højst skal vente på at 
modtage strengen, inden den næste kommando køres. Uden denne parameter venter computeren i al 
evighed.

Hvis computeren modtager en af de angivne strenge, indstilles systemvariablen $SUCCESS til TRUE. 
Den indstilles til FALSE, hvis det antal sekunder, der angives af tid, forløber, inden strengen modtages.  

Eksempler:

waitfor "Login:"

waitfor "Adgangskode?", matchcase

waitfor "prompt>" until 10

waitfor
"Login:" then DoLogin,
"Adgangskode:" then DoPassword,
"BBS:" then DoBBS,
"Andet:" then DoOther

 until 10

Følgende kommandoer er en Shiva-udvidelse til det kommandosprog for scriptfiler, der er udviklet 
af Microsoft.

getinput "Vis streng " szToReceiveInput

Beder brugeren om at indtaste oplysninger under udførelsen af scriptet.

Eksempel:

Getinput "System-adgangskode" szSystemPassword  



9.0   Reserverede ord
Følgende ord er reserverede og kan ikke bruges som variabelnavne:

and boolean databits delay
do endif endproc endwhile
even FALSE getip goto
halt if integer ipaddr
keyboard mark matchcase none
odd off on or
parity port proc raw
screen set space stopbits
string then transmit TRUE
until waitfor while

10.0 Shiva-kommandoord, som ikke understøttes af Scripter
getip
port databits
port parity
port stopbits
set screen keyboard
ipaddr
while/endwhile


